/*
Stan A. Kaplowitz	
Proposal Title: Examining How Message Characteristics Influence Support for Gasoline Tax Increases

HYPOTHESES

Stated-Hyp1: Support for increasing gasoline tax will increase if the message presents evidence that the tax will reduce gasoline consumption. (p. 4)

	Test-Hyp1: Support for increasing gasoline tax will be higher in the condition where the message has evidence that tax will reduce gasoline consumption vs. no evidence of consumption reduction.
	
Stated-Hyp2: Support for increasing gasoline tax will increase if the message presents evidence that the tax burden will not be excessive. (p. 4)
	
	Test-Hyp2: Support for increasing gasoline tax will be higher if the message presents evidence that the tax burden will not be excessive vs. control (phased tax increase).
	
	Test-Hyp3: Support for increasing gasoline tax will be higher if the message presents evidence that the tax burden will not be excessive vs. control (reimbursement through reinvestment).
	


Stated-Hyp3: Respondents who do not believe in anthropogenic climate change will have lower support for the tax. (p. 4)	

	Test-Hyp4: Support for tax will increase with increasing belief in anthropogenic climate change.		

Stated-Hyp4: Respondents who oppose high taxes will have lower support for the gasoline tax. (p. 4)		

	Test-Hyp5: Support for gasoline tax will decrease with increasing opposition to high taxes.

Stated-Hyp5: Respondents who do not believe in climate change will support the tax if they believe reducing oil consumption is good for America's economic security. (p. 4)	
	// Note: not clear how to test this. I'm creating a cleaner test by splitting both measures at the median

	Test-Hyp6: Among respondents who have below-median belief in climate change, support for the tax will increase with belief that consumption reduction benefits economic security.

Stated-Hyp6: Both of the following beliefs contribute to the belief reducing petroleum
consumption is desirable and also to acceptance of increasing the gasoline tax: (a) that human caused
climate change is a significant danger and (b) that America's national security is endangered by our
dependence on petroleum
	
	// cannot test. Measures were not collected. 

	
********************************************************************************
NOTES:

- Evidence that the burden will not be execessive is operationlized in two 
different treatments (phased tax and reimbursement) described on proposal page 3. 
So stated hypothesis 2 has two different test hypotheses.

- Stated hypotheses 1 and 2 only compare whether evidence was provided or not 
(i.e. there is no hypothesis about partial/less convincing vs. more convincing 
evidence). We have recoded these variables into dichotomous measures.

- Proposal mentions examining interactions by sociodemographic characters but 
does not specify variables  (p. 4). Not including any test because of rule about 
unspecified moderators.

- Some moderators outlined in the proposal are not included in survey. We are 
using all available ones.

- Beliefs about anthropogenic climate change: there are 3 questions related to 
this construct. 2 are measured on a scale, one has unordered outcome. Changing 
the unordered outcome to a dichotomous measure and adding that to a summated 
scale.

- For stated hypothesis 3, we are defining "don't believe in climate change" 
as below-median belief in anthropogenic climate change. 
	*/
	
clear all
use "kaplowitz160.dta", clear


********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATION

* evidence that tax burden will not be excessive: phased tax increase
	// var 1 in quex page 5
	recode XTESS3132 (1/9=0) (10/18=1), gen(phased)	
	tab phased 
	
* evidence that tax burden will not be excessive: tax revenues reinvested in transportation
	// var 2 described in quex page 5
	recode XTESS3132 (1 4 7 10 13 16=0 "0 control") (2 5 8 11 14 17=1 "1 tax refund") (3 6 9 12 15 18=2 "2 tax refund+ clean energy"), gen(reinvest)
	tab reinvest, mis	
	* Note: stated hypothesis 2 only compares any evidence of non-excessive burden vs. no evidence. Creating a dichotomous measure	
	recode reinvest (0=0) (1/2=1), gen(reinvest_binary)
	
* evidence that tax will enhance not be excessive: fuel efficiency + lower consumption
	// var 3 described in quex page 5
	recode XTESS3132 (1 2 3 10 11 12=0 "0 control") (4 5 6 13 14 15=1 "1 fuel eff.") (7 8 9 16 17 18=2 "2 fuel eff. + low cons."), gen(conserve)
	tab conserve	
	* Note: hypothesis 1 only compares control to low consumption condition. Creating a dichotomous measure
	recode conserve (0=0) (1/2=1), gen(conserve_binary)
		
	
	
* OUTCOME

* support for gasoline taxes
	*Q6. How much would you support or oppose the proposal for a $1 per gallon increase in the gasoline tax?
	
	tab Q6
	replace Q6=. if Q6<0
	gen support_tax= Q6
	
	
* MODERATORS

* belief about anthropogenic climate change
	/* NOTE: there are 3 questions related to this construct. 
	2 are measured on a scale, one has unordered outcome.
	I'm changing the unordered outcome to a dichotomous measure and adding that 
	to a summated scale
	*/

	*Q1. Which of the following statements BEST reflects your view of when 
		//significant effects of global warming will begin to happen?	
	tab Q1
	replace Q1=. if Q1<0	
	gen belief1 = Q1 
	
	*Q2. The scientific community agrees that the Earth’s temperature has 
		//increased over the last century.  What do you believe is the primary cause of this increase?
	tab Q2
	replace Q2=. if Q2<0
	recode Q2 (1 3=0) (2=1), gen(belief2)
	tab belief2


	*Q4. Superstorm Sandy in Fall 2012 was made more severe by the increased 
		//amount of greenhouse gasses that humans have put into the atmosphere.
	tab Q4
	replace Q4=. if Q4<0
	gen belief3 = 6-Q4 // originally reverse coded
	
	* create summated scale
	gen anthropogen_change= belief1+belief2+belief3
	tab anthropogen_change
		
	
* belief that reducing oil consumption is good for economic security
	*Q3. The health and economic well-being of America and the world require that we limit global warming.  This requires that Americans substantially reduce our use of oil.
	tab Q3
	replace Q3=. if Q3<0
	gen econsecurity = 6-Q3 // recode so higher values reflect agreement
	tab econsecurity
	
	
* opposition to high taxes
	* Q5. What is your view of the amount of taxes paid by people like you
	tab Q5
	replace Q5=. if Q5<0
	gen hightaxoppose= Q5
	tab hightaxoppose

********************************************************************************
* ANALYSIS

* Test-Hyp1: Support for the tax will be higher in the condition where the message has evidence that tax will reduce gasoline consumption vs. no evidence of consumption reduction.
	reg support_tax i.conserve_binary
	// reject. p=0.248
	tess 1.conserve_binary +, init(Kaplowitz160)
	
	
*Test-Hyp2: Support for increasing gasoline tax will be higher if the message presents evidence that the tax burden will not be excessive vs. control (phased tax increase).	
	reg support_tax i.phased
	// reject. p=0.090
	tess 1.phased +		
	
	
*Test-Hyp3: Support for increasing gasoline tax will be higher if the message presents evidence that the tax burden will not be excessive vs. control (reimbursement through reinvestment).	
	reg support_tax i.reinvest_binary
	// reject. p=0.496
	tess 1.reinvest_binary +		
	
	
* Test-Hyp4: Support for tax will increase with increasing belief in anthropogenic climate change.	
	reg support_tax anthropogen_change
	// do not reject. 0.000
	tess anthropogen_change +
	
* Test-Hyp5: Support for gasoline tax will decrease with increasing opposition to high taxes.
	reg support_tax hightaxoppose
	// do not reject. 0.000
	tess hightaxoppose -	
	
*Test-Hyp6: Among respondents who have below-median belief in climate change, support for the tax will increase with belief that consumption reduction benefits economic security.

	sum anthropogen_change, d
	reg support_tax econsecurity if anthropogen_change<`=r(p50)'
	// do not reject. p=0.000
	tess econsecurity +
